#
# QLogger library project file
#
TEMPLATE = lib
CONFIG += c++11

contains( CONFIG, staticlib ) {

}
else {
CONFIG += dll
}

QT -= gui

CONFIG(debug, debug|release) {
OBJECTS_DIR = build/debug
MOC_DIR     = build/debug
}
else {
OBJECTS_DIR = build/release
MOC_DIR     = build/release
}

#TODO: here in future replace
contains( CONFIG, dll ) {
    win32:DEFINES += QLOGGER_SHARED
}

include(src/qlog.pri)

CONFIG(debug, debug|release) {
    TARGET = qloggerd
} else {
    TARGET = qlogger
}

DESTDIR = ../lib/$$basename(QMAKESPEC)/$$QT_VERSION

target.path     += d:/tmp/AKRD/
INSTALLS        += target

OTHER_FILES += \
    ../TODO.txt

INCLUDEPATH += ../include
HEADERS += \
    ../include/qlogger.hpp



